Astronomical Computing

Conducted by Roger W. Sinnott

UPS AND DOWNS OF THE MOON

APART from the weather, nothing controls the nightly routines of astronomers so much as the Moon. Will the sky be clear at the observing site? How much observing can be done before moonrise? When will that be, anyway?

To calculate moonrise or moonset from scratch is a messy problem, far too tedious to do on a pocket calculator. Sooner or later everyone wants a canned computer program, reliable to a minute or two and valid anywhere in the world. For added versatility, the program should give correct answers for dates that are hundreds of years in the past or future.

By tradition, moonrise or moonset occurs when the upper limb of the Moon appears on the horizon at sea level. The calculation must take atmospheric refraction into account, of course, for it bends a grazing light ray "around" the horizon by 34 arc minutes under normal meteorological conditions. The Moon's distance is also needed, for it varies during the month and controls not only the size of the lunar disk but also the parallax effect at the horizon.

The rule adopted by the world's nautical almanacs is actually rather simple to state. For moonrise and moonset on a given date, we need to ask at what times the Moon's zenith distance equals 90° 34′ plus the semidiameter of the lunar disk minus the horizontal parallax. Answering that question is the task of this month's Basic program.

The program handles calendar dates, sidereal time, and sky coordinates in ways already discussed in this department (*S&T:* May, 1984, page 454, and June, 1984, page 558). At its heart, though, is a subroutine to compute the Moon's right ascension, declination, and distance by the method in T. C. Van Flandern and K. F. Pulkkinen's 1979 paper, "Low-Precision Formulae for Planetary Positions." Only enough periodic terms are retained to assure an accuracy of 0°.05, which corresponds to better than a minute in the calculated time of moonset or moonrise.

```
10 REM          MOONRISE-MOONSET
15 GOSUB 170
20 INPUT "LAT, LONG (DEG)";B5,L5
25 INPUT "TIME ZONE (HRS)";H
30 L5=L5/360: Z0=H/24
35 GOSUB 760: T=(J-2451545)+F
40 GOSUB 245: T=T+Z0
45 REM
50 REM          POSITION LOOP
55 FOR I=1 TO 3
60 GOSUB 495: M(I,1)=A5
65 M(I,2)=D5: M(I,3)=R5: T=T+0.5
70 NEXT
75 IF M(2,1)>M(1,1) THEN 85
80 M(2,1)=M(2,1)+P2
85 IF M(3,1)>M(2,1) THEN 95
90 M(3,1)=M(3,1)+P2
95 Z1=R1*(90.567-41.685/M(2,3))
100 S=SIN(B5*R1): C=COS(B5*R1)
105 Z=COS(Z1): M8=0: W8=0: PRINT
110 A0=M(1,1): D0=M(1,2)
115 FOR C0=0 TO 23
120 P=(C0+1)/24
125 F0=M(1,1):F1=M(2,1):F2=M(3,1)
130 GOSUB 225: A2=F
135 F0=M(1,2):F1=M(2,2):F2=M(3,2)
140 GOSUB 225: D2=F
145 GOSUB 285: A0=A2:D0=D2:V0=V2
150 NEXT
155 GOSUB 450: REM  SPECIAL MSG?
160 END
165 REM
170 REM          CONSTANTS
175 DIM M(3,3)
180 P1=3.14159265: P2=2*P1
185 R1=P1/180: K1=15*R1*1.0027379
190 S$="MOONSET AT  "
195 R$="MOONRISE AT "
200 M1$="NO MOONRISE THIS DATE"
205 M2$="NO MOONSET THIS DATE"
210 M3$="MOON DOWN ALL DAY"
215 M4$="MOON UP ALL DAY"
220 RETURN
225 REM     3-POINT INTERPOLATION
230 A=F1-F0: B=F2-F1-A
235 F=F0+P*(2*A+B*(2*P-1))
240 RETURN
245 REM       LST AT 0H ZONE TIME
250 T0=T/36525
255 S=24110.5+8640184.813*T0
260 S=S+86636.6*Z0+86400*L5
265 S=S/86400: S=S-INT(S)
270 T0=S*360*R1
275 RETURN
280 REM
285 REM  TEST AN HOUR FOR AN EVENT
290 L0=T0+C0*K1: L2=L0+K1
295 IF A2<A0 THEN A2=A2+2*P1
300 H0=L0-A0: H2=L2-A2
305 H1=(H2+H0)/2: REM  HOUR ANGLE
310 D1=(D2+D0)/2: REM  DEC
315 IF C0>0 THEN 325
320 V0=S*SIN(D0)+C*COS(D0)*COS(H0)-Z
325 V2=S*SIN(D2)+C*COS(D2)*COS(H2)-Z
330 IF SGN(V0)=SGN(V2) THEN 440
335 V1=S*SIN(D1)+C*COS(D1)*COS(H1)-Z
340 A=2*V2-4*V1+2*V0: B=4*V1-3*V0-V2
345 D=B*B-4*A*V0: IF D<0 THEN 440
350 D=SQR(D)
355 IF V0<0 AND V2>0 THEN PRINT R$;
360 IF V0<0 AND V2>0 THEN M8=1
365 IF V0>0 AND V2<0 THEN PRINT S$;
370 IF V0>0 AND V2<0 THEN W8=1
375 E=(-B+D)/(2*A)
380 IF E>1 OR E<0 THEN E=(-B-D)/(2*A)
385 T3=C0+E+1/120: REM  ROUND OFF
390 H3=INT(T3): M3=INT((T3-H3)*60)
395 PRINT USING "##:##";H3;M3;
400 H7=H0+E*(H2-H0)
405 N7=-COS(D1)*SIN(H7)
```

(Continued next page)

To photograph the April 6th young Moon, only 24 hours past new, Jerry Schad found a site overlooking the El Cajon Valley near San Diego, California. For this location near 33° north and 117° west, the program gives a moonset time of 20:19 PDT at azimuth 289°.

```
410 D7=C*SIN(D1)-S*COS(D1)*COS(H7)
415 A7=ATN(N7/D7)/R1
420 IF D7<0 THEN A7=A7+180
425 IF A7<0 THEN A7=A7+360
430 IF A7>360 THEN A7=A7-360
435 PRINT USING ",  AZ ###.#";A7
440 RETURN
445 REM
450 REM   SPECIAL MESSAGE ROUTINE
455 IF M8=0 AND W8=0 THEN 475
460 IF M8=0 THEN PRINT M1$
465 IF W8=0 THEN PRINT M2$
470 GOTO 485
475 IF V2<0 THEN PRINT M3$
480 IF V2>0 THEN PRINT M4$
485 RETURN
490 REM
495 REM   FUNDAMENTAL ARGUMENTS
500 L=0.606434+0.03660110129*T
505 M=0.374897+0.03629164709*T
510 F=0.259091+0.03674819520*T
515 D=0.827362+0.03386319198*T
520 N=0.347343-0.00014709391*T
525 G=0.993126+0.00273777850*T
530 L=L-INT(L): M=M-INT(M)
535 F=F-INT(F): D=D-INT(D)
540 N=N-INT(N): G=G-INT(G)
545 L=L*P2: M=M*P2: F=F*P2
550 D=D*P2: N=N*P2: G=G*P2
555 V=0.39558*SIN(F+N)
560 V=V+0.08200*SIN(F)
565 V=V+0.03257*SIN(M-F-N)
570 V=V+0.01092*SIN(M+F+N)
575 V=V+0.00666*SIN(M-F)
580 V=V-0.00644*SIN(M+F-2*D+N)
585 V=V-0.00331*SIN([illegible]2*D+N)
590 V=V-0.00304*SIN(F-2*D)
595 V=V-0.00240*SIN(M-F-2*D-N)
600 V=V+0.00226*SIN(M+F)
605 V=V-0.00108*SIN(M+F-2*D)
610 V=V-0.00079*SIN(F-N)
615 V=V+0.00078*SIN(F+2*D+N)
620 U=1-0.10828*COS(M)
625 U=U-0.01880*COS(M-2*D)
630 U=U-0.01479*COS(2*D)
635 U=U+0.00181*COS(2*M-2*D)
640 U=U-0.00147*COS(2*M)
645 U=U-0.00105*COS(2*D-G)
650 U=U-0.00075*COS(M-2*D+G)
655 W=0.10478*SIN(M)
660 W=W-0.04105*SIN(2*F+2*N)
665 W=W-0.02130*SIN(M-2*D)
670 W=W-0.01779*SIN(2*F+N)
675 W=W+0.01774*SIN(N)
680 W=W+0.00987*SIN(2*D)
685 W=W-0.00338*SIN(M-2*F-2*N)
690 W=W-0.00309*SIN(G)
695 W=W-0.00190*SIN(2*F)
700 W=W-0.00144*SIN(M+N)
705 W=W-0.00144*SIN(M-2*F-N)
710 W=W-0.00113*SIN(M+2*F+2*N)
715 W=W-0.00094*SIN(M-2*D+G)
720 W=W-0.00092*SIN(2*M-2*D)
725 REM
730 REM   COMPUTE RA, DEC, DIST
735 S=W/SQR(U-V*V)
740 A5=L+ATN(S/SQR(1-S*S))
745 S=V/SQR(U):D5=ATN(S/SQR(1-S*S))
750 R5=60.40974*SQR(U)
755 RETURN
760 REM     CALENDAR --> JD
765 INPUT "Y,M,D ";Y,M,D
770 G=1: IF Y<1582 THEN G=0
775 D1=INT(D): F=D-D1-0.5
780 J=-INT(7*(INT((M+9)/12)+Y)/4)
785 IF G=0 THEN 805
790 S=SGN(M-9): A=ABS(M-9)
795 J3=INT(Y+S*INT(A/7))
800 J3=-INT((INT(J3/100)+1)*3/4)
805 J=J+INT(275*M/9)+D1+G*J3
810 J=J+1721027+2*G+367*Y
815 IF F>=0 THEN 825
820 F=F+1: J=J-1
825 RETURN
```

(In the arctic regions, the uncertainty may reach several minutes.) The Van Flandern-Pulkkinen paper is available for $6 from Willmann-Bell, Inc., P. O. Box 35025, Richmond, Va. 23235.

PROGRAM OPERATION

The computer will first ask for your latitude, longitude, and time zone. North latitudes are considered positive, and west longitudes *negative,* in accord with current astronomical practice. The time zone is simply the number of hours you routinely add for converting civil to Universal time (see the paragraph on page 72). Finally you are asked for a year, month, and day. Enter the civil date on which you need the times of moonrise and moonset.*

For example, Los Angeles is at north latitude 34°.1, west longitude 118°.3, and keeps Pacific daylight time this summer. So folks there can enter "34.1, −118.3" and "7" in response to the questions. On July 4th of this year (enter "1989, 7, 4"), the program says moonrise will occur at 7:08 and moonset at 21:39 (9:39 p.m.) PDT. It also provides the azimuths of these events: 63° and 294°, respectively.

The azimuths are a bonus, mainly of interest to photographers wishing to record the low Moon behind a distant

*People who like to think in Universal time can enter 0 for the time zone. The 24-hour interval searched will be the UT date, and output times will be expressed in UT as well.

landmark on Earth. A topographic map and protractor are all you need in planning such a picturesque scene. Azimuths are counted around the horizon from north (0°) through east (90°), south (180°), and west (270°).

In the arctic regions the Moon can remain above or below the horizon all day, a phenomenon similar to the midnight Sun and the long winter night. The program will tell you as much. But another quirk of the arctic is not so well known — on certain days the Moon can rise or set not once, but twice!

Try the program using latitude +66°, longitude 0°, time zone 0, and June 23, 1989. You should find a moonrise at 0:04, moonset at 7:04, and then another moonrise at 23:46. The last event is real, even though it is missing from the 1989 *Astronomical Almanac*.

R. W. S.

Electronic News Service

Computer owners with access to modems can obtain quick news from *Sky & Telescope* of comet and nova discoveries, and also program listings from the Astronomical Computing department. After you join CompuServe (through a local Radio Shack or other computer dealer), log on and at any ! prompt type GO SKYTEL. This takes you directly to the S&T menu.

Review Corner

N-Body Solutions, Bartelheimer Scientific Software, Mid City Station Box 402, Dayton, Ohio 45402. $39.95 on two 5¼-inch disks or one 3½-inch disk for IBM PC computers; with manual.

Gravity, Cross Educational Software, 504 E. Kentucky Ave., P. O. Box 1536, Ruston, La. 71270. $49.00 on two 5¼-inch disks for IBM PC computers; with manual.

Gravitation Ltd., Jeff Rommereide, 343 Elma Ave., Laurel Springs, N. J. 08021. $20.00 on disk for Macintosh computers.

GRAVITY is the glue that holds the universe together. But it's not easy to visualize how it works when more than two bodies are involved. Voyager's path past Jupiter, Saturn, Uranus, and Neptune — while all five orbit the Sun — is dreadfully complex. These three programs show just how intricate and unexpected gravitational interactions can be.

All three programs let you invent gravitational systems, crank them up, and watch them run. You can pause in midstep, adjust masses or velocities, zoom in or out, see an instant replay, change the time interval, and in general create little solar systems that you control. Teachers should use one of these programs whenever gravity is taught. After manipulating your own worlds on screen, equations alone will never be enough.

Each program has strengths and weaknesses. *N-Body* uses real units (kilograms and seconds), so you can simulate real planets and moons — up to 34 at a time. You might enter data for Pluto and its moon, then send Voyager past to visualize how "it might have been." Changes, however, are somewhat laborious to make and you can spend as much time keying in long numbers as watching the display. And if you make a mistake, you are abruptly dumped out of the program. It assumes that you know what you're doing, and the manual was written for neither novices nor educators.

Gravity uses arbitrary units to specify mass and velocity but otherwise has options similar to *N-Body*'s. A major difference is that the program is better organized; you are continuously prompted and guided and thus unlikely to get lost or confused. This encourages experimentation. The program's author, Richard Palmaccio, is a teacher, and it shows.

Gravitation Ltd. is the only program of these three that shows the bodies as disks (rather than bare dots) whose sizes are proportional to their masses. This turns out to be tremendously important for visualizing what is going on. My four-year-old enjoys guessing the fates of the planets and moons: when two collide, a bigger one emerges on a new path. There is no manual, but the program is entirely intuitive. You can stop a simulation, use the "planet editor," and resume in seconds. It's fun. The two dozen examples, most of which are bizarre situations I never would have thought up, kept me busy for hours. With proper guidance, it could be used from lower elementary grades through university level.

My recommendation is to use *Gravity* if you're tied to the IBM world (unless you need to model a real situation), but to go for *Gravitation Ltd.* if you have access to a Macintosh.

JOHN MOSLEY
Griffith Observatory
2800 E. Observatory Rd.
Los Angeles, Calif. 90027

Will this planet's trio of moons eventually collide? Or will one be ejected? This simulation, called "moon on moon," comes with *Gravitation Ltd.* for the Macintosh.